Microsoft Access Visual Basic 函数

Command 函数


使用 Command 函数可以返回用以启动 Microsoft Access 的命令行 （命令行：用来启动应用程序的字符串。）的参数部分。

说明

当从命令行启动 Microsoft Access 时，命令行中跟在 /cmd 选项后的部分将会被作为命令行参数传给 Microsoft Access。使用 Command 函数可返回该参数。

若要在打开数据库之后更改命令行参数，可单击“工具”菜单上的“选项”。在“选项”对话框“高级”选项卡上的“命令行参数”框中输入新的参数。此时 Command 函数将返回新输入的参数。

在模块的 Visual Basic 代码之外使用 Command 函数时，必须在该函数后加上空的括号对。例如，若要在窗体上的文本框中使用 Command 函数，必须将文本框的 ControlSource 属性设为以下的表达式 （表达式：运算符、字段名、函数、文字和常量的组合，可得出单个结果。表达式可指定条件（例如，Order Amount>10000）或在字段值上执行计算（例如，Price*Quantity）。）：

=Command()

示例

下面的示例显示了如何使用带参数的命令行启动 Microsoft Access，以及如何使用 Command 函数返回该参数的值。

若要测试该示例，请单击 Windows 的“开始”按钮，然后单击“运行”，接着在“运行”对话框中按单行键入以下代码（必须将命令行信息部分用引号括起来）。

"C:\Program Files\Microsoft Office\Office10\Msaccess.exe" _

"C:\Program Files\Microsoft Office\Office10\Samples\Northwind.mdb" /cmd "Orders"

下一个示例是在“罗文斯”示例数据库中新建模块并添加以下 Sub 函数。

Public Sub CheckCommandLine()

    ' Check the value returned by Command function and display

    ' the appropriate form.

    If Command = "Orders" Then

        DoCmd.OpenForm "Orders"

    ElseIf Command = "Employees" Then

        DoCmd.OpenForm "Employees"

    Else

        Exit Sub

    End If

End Sub

调用该过程时，Microsoft Access 将打开“订单”窗体。在打开数据库时可创建 AutoExec 宏来调用该过程。

DAvg 函数

参阅应用于示例特性

使用 DAvg 函数可以计算特定记录集（一个域 （域：由表、查询或 SQL 表达式定义的记录集。域聚合函数返回有关特定域或记录集的统计信息。））内一组值的平均值。可以在 Visual Basic 代码、宏 （宏：可用来使任务自动化的操作或操作集。宏在 Visual Basic for Applications 编程语言中录制。）、查询表达式或计算控件 （计算控件：在窗体、报表或数据访问页上用来显示表达式结果的控件。每当表达式所基于的值发生改变，就重新计算一次结果。）中使用 DAvg 函数。

例如，可以在选择查询的运费字段所对应的条件行中使用 DAvg 函数，来限制返回运费超过平均值以上的记录。或者在计算控件中使用包含 DAvg 函数的表达式，在新订单值旁边显示旧订单的平均值。

DAvg(expr, domain, [criteria])

DAvg 函数具有以下参数：

	参数
	说明

	expr
	一个表达式，代表要计算其平均值的数值数据字段。可以是用来标识表或查询中字段的字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），也可以是对该字段上的数据执行计算的表达式。在 expr 中可以包括表中字段的名称、窗体上的控件、常量或函数。如果 expr 包含函数，那么它可以是内置的，也可以是用户定义的，但不能是另一个域聚合函数或 SQL 聚合函数。

	domain
	字符串表达式，代表组成该域的记录集。可以是表名称或不需要参数的查询名称。

	criteria
	可选的字符串表达式，用于限制 Davg 函数执行的数据范围。例如，criteria 通常等价于 SQL 表达式中的 WHERE 子句，只是不含 WHERE 关键字。如果忽略 criteria，DAvg 函数将对整个域计算 expr。任何包含在 criteria 中的字段必须同时也是 domain 中的字段，否则 DAvg 函数将返回 Null （Null：可以在字段中输入或用于表达式和查询，以标明丢失或未知的数据。在 Visual Basic 中，Null 关键字表示 Null 值。有些字段（如主键字段）不可以包含 Null 值。）。


说明

包含 Null 值的记录不能在平均值的计算中使用。

无论是在宏、模块、查询表达式还是在计算控件中使用 DAvg 函数，都必须仔细地构造 criteria 参数，以确保能够正确地进行计算。

在查询的“条件”行中可以使用 DAvg 函数来指定条件。例如，要查看所有定购数量在平均订购量以上的产品列表，可以基于“Orders”、“Order Details”和“Products”表创建一个查询，包含“Product Name”和“Quantity”字段，并在“Quantity”字段下的“条件”行中加入以下表达式：

>DAvg("[Quantity]", "Orders")

在查询的计算字段表达式中，或更新查询 （更新查询：一种操作查询（SQL 语句），它根据所指定的条件（查找条件）更改记录集。）中“更新到”行中的计算字段表达式中，也可以使用 DAvg 函数。

注释  在总计查询 （总计查询：这种查询可针对表（一个或多个）中各不同字段显示汇总计算，如平均值或总计值。总计查询不是一个单独种类的查询，而是扩展了选择查询的灵活性。）的计算字段表达式中，既可以使用 DAvg 函数，也可以使用 Avg 函数。如果使用 DAvg 函数，将在数据分组之前计算平均值。如果使用 DAvg 函数，则在数据分组之后计算字段表达式的平均值。

当需要指定条件来限制 DAvg 函数执行的数据范围时，应在计算控件中使用 DAvg 函数。例如，要显示到 California 的运费的平均值，应将文本框的 ControlSource 属性设为如下表达式：

=DAvg("[Freight]", "Orders", "[ShipRegion] = 'CA'")

如果只对所有 domain 中的记录计算平均值，则可使用 Avg 函数。

如果要显示的字段不在窗体的基础记录源中时，可以在模块或宏中，或者窗体上的计算控件中，使用 DAvg 函数。例如，有一个基于“订单”表的窗体，希望包含“订单明细”表中的“数量”字段，以显示按特定客户排序的项目平均值，可以使用 DAvg 函数来执行这项计算并在窗体中显示该数据。

提示
· 在计算控件中使用 DAvg 函数时，有时需要将控件放在窗体页眉或页脚中，以便该控件的值在每次移动到新记录时不必重新计算。 

· 如果 expr 所源于的字段数据类型为数字，则 DAvg 函数将返回 Double （Double 数据类型：一种基础数据类型，存储双精度浮点数。它存储 64 位数，范围大约介于 -1.797E308 到 -4.940E-324（负）之间、4.94E-324 到 1.797E308（正）之间和 0。） 数据类型。如果在计算控件中使用 DAvg 函数，可在表达式中包含数据类型转换函数，以提高性能。 

· 虽然使用 DAvg 函数可以确定外部表 （外表：包含外键字段（如“客户ID”）的表，如“客户订单”。外键字段在其他表（例如“客户”）中是主键字段，并且通常处于一对多关系的“多”端。）字段的数据平均值，但是通过创建含有所有需要字段的查询，然后将窗体或报表建立在这个查询的基础上，效率将更高。 

注释  当使用此函数时，不包括对 domain 记录未保存的更改。如果希望 DAvg 函数基于更改后的值，必须首先保存更改，方法是：单击“记录”菜单中的“保存记录”、将焦点移动到另一条记录，或使用 Update 方法。

示例

下列函数返回特定日期或特定日期以后已发订货的平均运费。域为“Orders”表。条件参数根据给定的国家/地区和发货日期来限制生成的记录集。请注意，关键字 AND 包含在字符串内，用于分隔条件参数内的多个字段。包含在 DAvg 函数计算内的所有记录都将满足这两个条件。

Public Function AvgFreightCost(ByVal strCountry As String, _

                               ByVal dteShipDate As Date) As Double

    AvgFreightCost = DAvg("[Freight]", "Orders", _

                     "[ShipCountry] = '" & strCountry & _

                     "'AND [ShippedDate] >= #" & dteShipDate & "#")
End Function

若要调用该函数，请在“立即”窗口中使用下列代码行：

:AvgFreightCost "UK", #1/1/96#

DCount 函数

参阅应用于示例特性

使用 Dcount 函数可以确定特定记录集（一个域 （域：由表、查询或 SQL 表达式定义的记录集。域聚合函数返回有关特定域或记录集的统计信息。））中的记录数。可以在 Visual Basic、宏 （宏：可用来使任务自动化的操作或操作集。宏在 Visual Basic for Applications 编程语言中录制。）、查询表达式，或计算控件 （计算控件：在窗体、报表或数据访问页上用来显示表达式结果的控件。每当表达式所基于的值发生改变，就重新计算一次结果。）中使用 DCount 函数。

例如，可以在模块中使用 Dcount 函数，以返回“订单”表中对应于特定日期的订单数。

DCount(expr, domain, [criteria])

DCount 函数具有下列参数：

	参数
	说明

	expr
	一个表达式，代表要统计其记录数的字段。可以是标识表或查询中字段的字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），也可以是对该字段上的数据进行计算的表达式。在 expr 中可以包括表中字段的名称、窗体上的控件、常量或函数。如果 expr 包括函数，它既可以是内置的，也可以是用户定义的，但不能是另一个域聚合函数或 SQL 聚合函数。

	domain
	字符串表达式，代表组成域的记录集。可以是表名称或不需要参数的查询名称。

	criteria
	可选的字符串表达式，用于限制 DCount 函数执行的数据范围。例如，criteria 通常等价于 SQL 表达式中的 WHERE 子句，只是不含 WHERE 关键字。如果忽略 criteria，DCount 函数在整个域的范围内计算 expr。任何包含在 criteria 中的字段必须同时也是 domain 中的字段；否则 Dcount 函数将返回 Null （Null：可以在字段中输入或用于表达式和查询，以标明丢失或未知的数据。在 Visual Basic 中，Null 关键字表示 Null 值。有些字段（如主键字段）不可以包含 Null 值。）。


说明

在不需要知道记录特定值的时候，可以使用 DCount 函数来计算域中的记录数。虽然 expr 参数可以对字段执行计算，但 DCount 函数只返回记录数。任何由 expr 执行计算所得的值都是不可用的。

当需指定条件来限制函数执行的数据范围时，应在计算控件中使用 DCount 函数。例如，要显示运往 California 的订单数，可将文本框的 ControlSource 属性设为如下表达式：

=DCount("[OrderID]", "Orders", "[ShipRegion] = 'CA'")

如果只是统计 domain 中的所有记录而不指定任何限制，则可以使用 Count 函数。


提示

Count 函数是经过优化的，可以加快在查询中对记录数进行统计的速度。所以应在查询表达式中使用 Count 函数来代替 DCount 函数，并设置可选的条件以实施对结果的限制。如果必须从代码模块、宏或计算控件中对域中的记录进行计数时，应该使用 DCount 函数。

使用 DCount 函数可以计算包含特定字段的记录数，该字段不在窗体或报表所基于的记录源中。例如，在基于“产品”表的窗体中，其计算控件中可以显示“订单”表中的订单数。

DCount 函数在计数记录时，不包括由 expr 引用的字段中含有 Null 值的记录，除非 expr 参数为星号（*）通配符。如果使用了星号，DCount 函数将计算所有记录的总数，包括含有 Null 字段的记录。下面的示例将计算“Orders”表中的记录数。

intX = DCount("*", "Orders")

如果 domain 是一个包含主键 （主键：具有唯一标识表中每条记录的值的一个或多个域（列）。主键不允许为 Null，并且必须始终具有唯一索引。主键用来将表与其他表中的外键相关联。）的表，也可以将 expr 设为主键字段来计算总记录数，因为主键字段不可能为 Null。

如果 expr 指定多个字段，应使用连接运算符分隔各字段名，即和字符（&）或加号运算符（+）。如果使用（&）分隔字段，Dcount 函数将返回在列出的任一字段中含有数据的记录数。如果使用加号运算符，Dcount 函数将只返回在所有列出字段中都含有数据的记录数。下面的示例说明了使用这两种运算符的结果，它们连接所有记录中均包含数据的字段（ShipName）和不包含数据的字段（ShipRegion）。

intW = DCount("[ShipName]", "Orders")

intX = DCount("[ShipRegion]", "Orders")

intY = DCount("[ShipName] + [ShipRegion]", "Orders")

intZ = DCount("[ShipName] & [ShipRegion]", "Orders")

注释   和字符（&）在执行字符串连接时是首选的运算符。在除加法以外的其他情况下，应尽量避免使用加法运算符，除非特地需要在表达式中产生 Null 值。

使用该函数时，domain 中未保存更改的记录将不包括在内。如果要让 DCount 函数以更改后的数据为基础，必须先保存更改。方法是单击“记录”菜单上的“保存记录”，或是将焦点移动到另一个控件，或使用 Update 方法。

示例

下面的函数返回在发货日期之后发送到指定国家/地区的订单数量。域为“Orders”表。

Public Function OrdersCount(ByVal strCountry As String, _

                            ByVal dteShipDate As Date) As Integer

    OrdersCount = DCount("[ShippedDate]", "Orders", _

                  "[ShipCountry] = '" & strCountry & _

                  "' AND [ShippedDate] > #" & dteShipDate & "#")
End Function

若要调用该函数，请在“立即”窗口中使用下列代码：

:OrdersCount "UK", #1/1/96#

DDE 函数

参阅应用于示例特性

可以使用 DDE 函数来启动与另一个应用程序的动态数据交换 （动态数据交换 (DDE)：已建立的一种协议，用于在基于 Microsoft Windows 的程序之间交换数据。） (DDE) 对话，从该应用程序中请求某一信息，以及在窗体或报表上的控件中显示该信息。

例如，可以在文本框的“控件来源”属性中使用 DDE 函数来显示 Microsoft Excel 电子表格中指定单元格的数据。

DDE(application, topic, item)
DDE 函数具有以下参数：

	参数
	说明

	application
	字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），用于标识能参与 DDE 对话的应用程序。通常，application 是一个基于 Microsoft Windows 应用程序（如 Microsoft Excel）的 .exe 文件名（不带 .exe 扩展名）。例如，要启动一个与 Microsoft Excel 之间的 DDE 对话，应键入 "Excel" 作为 application 参数。

	topic
	字符串表达式，它是由 application 识别的主题 （主题：指两个应用程序之间动态数据交换 (DDE) 会话的主题。对于绝大多数使用文件的应用程序，主题就是文件名。）名称。该 topic 参数通常是文档或数据文件。请参考另一应用程序的文档以获得可用主题的列表。

	item
	字符串表达式，它是由 application 识别的数据项 （数据项：可以通过 DDE（动态数据交换）通道传送的应用程序专用数据。）名称。请参考另一应用程序的文档以获得可用项目的列表。


说明

DDE 函数用指定的 application 和 topic 来启动 DDE 对话，并请求 item 中的数据。如果成功，DDE 函数将返回包含请求信息的字符串。

如果要请求从 Microsoft Excel 中获取数据，item 必须是行与列的标识符，如 "R1C1" 或一个单元格区域的名称。在下面的示例中，该 DDE 函数请求从 Microsoft Excel 工作表中第 1 行、第 1 列的单元格中获取信息。在控件的属性表中，可以在“控件来源”属性框中为文本框控件输入下面的表达式。

=DDE("Excel", "Sheet1", "R1C1")

只能在文本框、选项组、复选框或组合框中的“控件来源”属性中使用 DDE 函数。不能从 Visual Basic 中调用 DDE 函数。

使用 DDE 函数时，在“窗体”视图 （“窗体”视图：一个显示窗体以便显示或接受数据的窗口。“窗体”视图是添加和修改表中数据的主要方式。在该视图中还可以更改窗体的设计。）和“打印预览” （打印预览：打印文档时显示文档的一种视图。）中，控件为只读。例如，如果在文本框中使用 DDE 函数，则不能编辑文本框中的文本。用户必须在另一应用程序中编辑文本。因为“控件来源”属性在“窗体”视图和“打印预览”中是只读的，对控件的更改必须在“设计”视图 （“设计”视图：显示数据库对象（包括：表、查询、窗体、宏和数据访问页）的设计的窗口。在“设计”视图中，可以新建数据库对象和修改现有数据库对象的设计。）中进行。

Microsoft Windows 和计算机的内存及资源决定了可以同时打开的 DDE 对话的最大数目。如果对话因为另一应用程序不在运行或不能识别指定主题而不能被启动，或已经达到最大的对话数目，DDE 函数则会返回 Null （Null：可以在字段中输入或用于表达式和查询，以标明丢失或未知的数据。在 Visual Basic 中，Null 关键字表示 Null 值。有些字段（如主键字段）不可以包含 Null 值。）。

注释  可以配置另一个应用程序忽略对 DDE 对话的请求。在这种情况下，DDE 函数将返回一个 Null 值。同样，也可以设置 Microsoft Access 忽略来自另一应用程序的请求：单击“工具”菜单上的“选项”，然后在“选项”对话框中“高级”选项卡上“DDE 操作”的下面，单击“忽略 DDE 请求”。


提示

如果需要从 Microsoft Access 中处理另一应用程序的对象，可以考虑使用“自动化”。

下表说明了对每对控件 （控件：允许用户控制程序的图形用户界面对象，如文本框、复选框、滚动条或命令按钮等。可使用控件显示数据或选项、执行操作或使用户界面更易阅读。）使用 DDE 函数时函数的行为。

	控件
	说明

	文本框
	item 参数能引用文本或数字。如果 item 引用多条信息（如 Microsoft Excel 工作表中包含多个单元格的命名区域），则 DDE 函数返回第一项。对文本框使用这个函数可以显示工作表上的单元格中包含的数据。

	组合框
	DDE 函数将 item 引用的信息填入组合框。不能在组合框的文本部分输入数据。将 DDE 函数用于组合框可以显示 Microsoft Excel 工作表中的区域或国家的列表。

	选项组
	选项组中每个选项按钮的“选项值”属性都设为一个数字。通常，第一个按钮的值为 1，第二个为 2，依次类推。DDE 函数所返回的数字将确定哪个选项按钮被选定。 

例如，如果 DDE 函数返回 2，则第二个按钮将被选定。如果该函数的返回值与“选项值”属性的任何设置都不匹配，则没有任何按钮被选定。如果 item 引用多条信息，如在 Microsoft Excel 工作表中包含多个单元格的命名区域，DDE 函数将返回第一项。

	复选框
	如果 DDE 函数返回 0，复选框将被清除。如果该函数返回非零值，如 1 或 -1，则复选框将被选中。如果 item 引用的是文本或多条信息，如 Microsoft Excel 工作表中包含多个单元格的命名区域，复选框将不可用。



[image: image6.png]


DDEInitiate 函数

参阅应用于示例特性

使用 DDEInitiate 函数可以开始与另一个应用程序的动态数据交换 (DDE) （动态数据交换 (DDE)：已建立的一种协议，用于在基于 Microsoft Windows 的程序之间交换数据。） 对话。DDEInitiate 函数打开 DDE 通道 （动态数据交换 (DDE) 通道：Microsoft Windows 程序之间的活动链接，通过该链接可交换数据。）以便在 DDE 服务器和客户应用程序之间传送数据。

例如，如果要将数据从 Microsoft Excel 电子表格传送到 Microsoft Access 数据库，可以使用 DDEInitiate 函数在这两个应用程序之间打开一条通道。在该示例中，Microsoft Access 充当客户应用程序，Microsoft Excel 充当服务器应用程序。

DDEInitiate(application, topic)

DDEInitiate 函数具有以下参数：

	参数
	说明

	application
	字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），标识可以参加 DDE 对话的应用程序。通常， application 参数是一个基于 Microsoft Windows 应用程序的 .exe 文件名（不带 .exe 扩展名），比如 Microsoft Excel。

	topic
	字符串表达式，它是能由 application 参数识别的主题 （主题：指两个应用程序之间动态数据交换 (DDE) 会话的主题。对于绝大多数使用文件的应用程序，主题就是文件名。）的名称。要想获得主题列表，请参考应用程序的文档说明。


说明

如果操作成功，则 DDEInitiate 函数用由 application 和 topic 参数指定的应用程序和主题来开始 DDE 对话，然后返回一个 Long （Long 数据类型：存储大型整数的基础数据类型。Long 变量存储为 32 位数的值，范围介于 -2,147,483,648 到 2,147,483,647 之间。） 整型值。该返回值代表一个唯一的通道号 （通道号：与打开的动态数据交换 (DDE) 通道相对应的整数。通道号由 Microsoft Windows 95 或更高版本指定，用 DDEInitiate 函数创建，并由其他 DDE 函数和语句使用。），标识可以传送数据的通道。该通道号随后即可和其他 DDE 函数及语句一起使用。

如果应用程序尚未运行或者虽在运行但不识别 topic 参数或不支持 DDE，则 DDEInitiate 函数将会返回一个运行时错误 （运行时错误：只有当应用程序运行时才能检测到的错误。）。

topic 参数的值取决于由 application 参数指定的应用程序。对于使用文档或数据文件的应用程序，有效的主题名通常包括这些文件的名称。

注释  可以同时打开的最大通道数由 Microsoft Windows 以及计算机的内存和资源决定。如果不打算使用通道，则应通过 DDETerminate 或 DDETerminateAll 语句将其终止以节省资源。


提示

如果需要从 Microsoft Access 中处理另一应用程序的对象，可以考虑使用“自动化”。

DDERequest 函数

参阅应用于示例特性

在一个打开的动态数据交换（DDE） （动态数据交换 (DDE)：已建立的一种协议，用于在基于 Microsoft Windows 的程序之间交换数据。）通道上使用 DDERequest 函数，可以从 DDE 服务器应用程序中请求一条信息。

例如，如果在 Microsoft Access 和 Microsoft Excel 之间有一条打开的 DDE 通道 （动态数据交换 (DDE) 通道：Microsoft Windows 程序之间的活动链接，通过该链接可交换数据。），则使用 DDERequest 函数可以将文本 Microsoft Excel 电子表格中传送给 Microsoft Access 数据库。

DDERequest(channum, item)

DDERequest 函数具有以下参数：

	参数
	说明

	channum
	通道号 （通道号：与打开的动态数据交换 (DDE) 通道相对应的整数。通道号由 Microsoft Windows 95 或更高版本指定，用 DDEInitiate 函数创建，并由其他 DDE 函数和语句使用。），由 DDEInitiate 函数返回的整型值。

	item
	字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），它是由 DDEInitiate 函数所指定的应用程序识别的数据项 （数据项：可以通过 DDE（动态数据交换）通道传送的应用程序专用数据。）名称。要想获得可能项的列表，请参考应用程序的文档。


说明

channum 参数指定所需的 DDE 对话的通道号，item 参数标识应从服务器应用程序中检索哪些数据。item 参数的值取决于在由 channum 参数指示的通道打开时指定的应用程序和主题 （主题：指两个应用程序之间动态数据交换 (DDE) 会话的主题。对于绝大多数使用文件的应用程序，主题就是文件名。）。例如，item 参数可能是 Microsoft Excel 电子表格中的单元格范围。

如果请求成功，DDERequest 函数将返回一个 Variant （Variant 数据类型：变量的默认数据类型，如果 Def type 语句没有生效，则该数据类型没有类型声明字符。Variant 可以存储数值、字符串、日期/时间、Null 或 Empty 数据。） 值作为包含请求信息的字符串 （字符串：字符集合，能包含数字和文字。）。

所请求的数据应是字母数字文本格式。其他任何格式的图形或文本都不能传送。

如果 channum 参数不是一个对应于打开的通道的整型值，或者如果所请求的数据无法传送，则会出现运行时错误 （运行时错误：只有当应用程序运行时才能检测到的错误。）。


提示

如果需要从 Microsoft Access 中处理另一应用程序的对象，可以考虑使用“自动化”。

DFirst、DLast 函数

参阅应用于示例特性

只需要字段中的任意一个值时，可以使用 DFirst 和 DLast 函数，从表或查询的特定字段中返回一个随机记录。可以在 Visual Basic、宏 （宏：可用来使任务自动化的操作或操作集。宏在 Visual Basic for Applications 编程语言中录制。）、模块、查询表达式或窗体或报表上的计算控件 （计算控件：在窗体、报表或数据访问页上用来显示表达式结果的控件。每当表达式所基于的值发生改变，就重新计算一次结果。）中使用 DFirst 和 DLast 函数。

DFirst(expr, domain, [criteria])

DLast(expr, domain, [criteria])

DFirst 和 DLast 函数具有下列参数：

	参数
	说明

	expr
	一个表达式，用于标识从中查找第一个或最后一个值的字段。可以是一个用来标识表或查询中字段的字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），也可以是以该字段中的数据进行计算的表达式。在 expr 中，可以包括表中字段的名称、窗体上的控件、常量或函数。如果 expr 包括函数，它既可以是内置的，也可以是用户定义的，但不能是另一个域聚合函数或 SQL 聚合函数。

	domain
	字符串表达式，代表组成域的记录集。

	criteria
	可选的字符串表达式，用于限制 DFirst 或 DLast 函数执行的数据范围。例如，criteria 通常等价于 SQL 表达式中的 WHERE 子句，只是不含 WHERE 关键字。如果忽略 criteria，DFirst 和 DLast 函数将在整个域范围内计算 expr。任何包含在 criteria 中的字段必须同时也是 domain 中的字段，否则 DFirst 和 Dlast 函数将返回 Null （Null：可以在字段中输入或用于表达式和查询，以标明丢失或未知的数据。在 Visual Basic 中，Null 关键字表示 Null 值。有些字段（如主键字段）不可以包含 Null 值。）。


说明

注释   如果要返回记录集（一个域 （域：由表、查询或 SQL 表达式定义的记录集。域聚合函数返回有关特定域或记录集的统计信息。））中的第一条或最后一条记录，应该创建一个按升序或降序排序的查询，并将 TopValues 属性设为 1。有关详细信息，请参阅 TopValues 属性主题。也可以从 Visual Basic 中创建 ADO Recordset 对象，并使用 MoveFirst 或 MoveLast 方法来返回记录集内的第一条或最后一条记录。

DLookup 函数

参阅应用于示例特性

DLookup 函数用于从指定记录集（一个域 （域：由表、查询或 SQL 表达式定义的记录集。域聚合函数返回有关特定域或记录集的统计信息。））获取特定字段的值。可以在 Visual Basic、宏 （宏：可用来使任务自动化的操作或操作集。宏在 Visual Basic for Applications 编程语言中录制。）、查询表达式、窗体或报表上的计算控件 （计算控件：在窗体、报表或数据访问页上用来显示表达式结果的控件。每当表达式所基于的值发生改变，就重新计算一次结果。）中使用 DLookup 函数。

使用 Dlookup 函数可以显示不在窗体或报表记录源中的字段值。例如，假定有一个基于“订单明细”表的窗体，显示“订单ID”、“产品ID”、“单价”、“数量”和“折扣”字段。而“产品名称”字段位于另一个表 （“产品”表）中。通过在计算控件中使用 Dlookup 函数，可以在同一窗体上显示“产品名称”。

DLookup(expr, domain, [criteria])

DLookup 函数具有下列参数：

	参数
	说明

	expr
	一个表达式，用于标识需要返回其值的字段。它既可以是用于标识表或查询中字段的字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），也可以是以该字段上的数据进行计算的表达式。在 expr 中，可以包括表中字段的名称、窗体上的控件、常量或函数。如果 expr 包含函数，那么它可以是内置的，也可以是用户定义的，但不能是另一个域聚合函数或 SQL 聚合函数。

	domain
	字符串表达式，代表组成域的记录集。可以是表名称或不需要参数的查询名称。

	criteria
	可选的字符串表达式，用于限制 DLookup 函数执行的数据范围。例如，criteria 通常等价于 SQL 表达式中的 WHERE 子句，只是不含 WHERE 关键字。如果忽略 criteria，DLookup 函数将在整个域范围内计算 expr。任何包含在 criteria 中的字段必须同时也是 domain 中的字段，否则 Dlookup 函数将返回 Null （Null：可以在字段中输入或用于表达式和查询，以标明丢失或未知的数据。在 Visual Basic 中，Null 关键字表示 Null 值。有些字段（如主键字段）不可以包含 Null 值。）。


说明

DLookup 函数将基于 criteria 中指定的信息返回单个字段的值。虽然 criteria 是可选参数，但如果不给 criteria 提供值，Dlookup 函数将返回域中的一个随机值。

如果没有记录满足 criteria，或者 domain 中没有记录，DLookup 函数将返回 Null。

如果有多个字段满足 criteria，DLookup 函数将返回第一个匹配字段。所以应该指定条件以确保 DLookup 函数返回的字段值是唯一的。可以在条件中使用主键 （主键：具有唯一标识表中每条记录的值的一个或多个域（列）。主键不允许为 Null，并且必须始终具有唯一索引。主键用来将表与其他表中的外键相关联。）值，在下例中，使用 [EmployeeID] 来确保 DLookup 函数返回唯一的值：

Dim varX As Variant

varX = DLookup("[LastName]", "Employees", "[EmployeeID] = 1")

无论是在宏、模块、查询表达式中，还是在计算控件中使用 DLookup 函数，都必须仔细地构造 criteria 参数，以确保能够正确地进行计算。

在查询的“条件”行、查询的计算字段表达式中或更新查询 （更新查询：一种操作查询（SQL 语句），它根据所指定的条件（查找条件）更改记录集。）的“更新到”行中均可以使用 DLookup 函数来指定条件。

如果需要显示的字段不在窗体或报表所基于的记录源中，也可以在窗体或报表的计算控件表达式中使用 DLookup 函数。例如，假定有一个“Order Details”窗体基于“Order Details”表，并且含有一个显示“ProductID”字段的名为“ProductID”的文本框，要从基于这个文本框值的“Products”表中查阅“ProductName”字段，可以创建另一个文本框，并将它的 ControlSource 属性设为如下表达式：

=DLookup("[ProductName]", "Products", "[ProductID] =" _

     & Forms![Order Details]!ProductID)

提示
· 虽然可以使用 DLookup 函数显示来自外部表 （外表：包含外键字段（如“客户ID”）的表，如“客户订单”。外键字段在其他表（例如“客户”）中是主键字段，并且通常处于一对多关系的“多”端。）字段中的值，但是通过创建包含两表中所需字段的查询，然后将窗体或报表建立在这个查询的基础上，效率将更高。 

· 也可以使用“查阅向导”来查找外部表中的值。 

注释  使用此函数时，不包括对 domain 记录未保存的更改。如果希望 DLookup 函数基于更改后的值，必须先保存更改。方法是：单击“记录”菜单上的“保存记录”命令、将焦点移到另一个记录上，或使用 Update 方法。

示例

下面的示例将从满足 criteria 内容记录的“CompanyName”字段中返回名称信息。域为“Shippers”表。criteria 参数将结果记录集限制为“ShipperID”等于 1 的记录。

Dim varX As Variant

varX = DLookup("[CompanyName]", "Shippers", "[ShipperID] = 1")
下一个示例来自 Shippers 表，它使用窗体控件 ShipperID 来为 DLookup 函数提供条件。请注意，控件的两侧未加上用以表示字符串的引号，这样可以确保每次调用 DLookup 函数时，Microsoft Access 都将从控件中获取当前值。

Dim varX As Variant

varX = DLookup("[CompanyName]", "Shippers", "[ShipperID] = " _

    & Forms!Shippers!ShipperID)
下一个示例使用变量 intSearch 来获取值。

Dim intSearch As Integer

Dim varX As Variant

intSearch = 1

varX = DLookup("[CompanyName]", "Shippers", _

    "[ShipperID] = " & intSearch)

DMin、DMax 函数

参阅应用于示例特性

DMin 和 DMax 函数用于确定指定记录集（一个域 （域：由表、查询或 SQL 表达式定义的记录集。域聚合函数返回有关特定域或记录集的统计信息。）) 中的最小值和最大值。可以在 Visual Basic、宏 （宏：可用来使任务自动化的操作或操作集。宏在 Visual Basic for Applications 编程语言中录制。）、查询表达式或计算控件 （计算控件：在窗体、报表或数据访问页上用来显示表达式结果的控件。每当表达式所基于的值发生改变，就重新计算一次结果。）中使用 DMin 和 DMax 函数。

例如，可以在报表上的计算控件中使用 DMin 和 DMax 函数来显示一个特定客户的最大和最小订单量。也可以在查询表达式中使用 DMin 函数显示折扣大于最小可能折扣的所有订单。

DMin(expr, domain, [criteria])

DMax(expr, domain, [criteria])

DMin 和 DMax 函数具有以下参数：

	参数
	说明

	expr
	表达式，用于标识要为其查找最小值和最大值的字段。也可以是用来标识表或查询中字段的字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），或者是以该字段中的数据进行计算的表达式。在 expr 中可以包括表字段的名称、窗体上的控件、常量或函数。如果 expr 含有函数，那么它可以是内置的，也可以是用户自定义的，但不能是另一个域聚合函数或 SQL 聚合函数。

	domain
	字符串表达式，代表组成域的记录集。可以是表名称或不需要参数的查询的查询名称。

	criteria
	可选的字符串表达式，用于限制 DMin 和 DMax 函数执行的数据范围。例如，criteria 通常等价于 SQL 表达式中的 WHERE 子句，只是不含 WHERE 关键字。如果忽略 criteria，DMin 和 DMax 函数将在整个域范围内计算 expr。任何包含在 criteria 中的字段必须同时也是 domain 中的字段，否则 DMin 和 Dmax 函数将返回 Null （Null：可以在字段中输入或用于表达式和查询，以标明丢失或未知的数据。在 Visual Basic 中，Null 关键字表示 Null 值。有些字段（如主键字段）不可以包含 Null 值。）。


说明

DMin 和 DMax 函数将返回满足 criteria 的最小值和最大值。如果 expr 标识的是数值数据，DMin 和 DMax 函数将返回数字值。如果 expr 标识的是字符串数据，则返回按字母顺序排列的第一个或最后一个字符串。

DMin 和 DMax 函数将忽略 expr 引用的字段中的 Null 值。但是，如果没有记录满足 criteria 或者 domain 中不包含记录，DMin 和 DMax 函数将返回 Null。

无论是在宏、模块、查询表达式或者在计算控件中使用 DMin 或 DMax 函数，都必须谨慎构造 criteria 参数以确保能够正确地进行计算。

在查询的“条件”行、查询的计算字段表达式中或更新查询 （更新查询：一种操作查询（SQL 语句），它根据所指定的条件（查找条件）更改记录集。）的“更新到”行中，都可以使用 DMin 和 DMax 函数指定条件。

注释  可以在总计查询 （总计查询：这种查询可针对表（一个或多个）中各不同字段显示汇总计算，如平均值或总计值。总计查询不是一个单独种类的查询，而是扩展了选择查询的灵活性。）的计算字段表达式中使用 DMin 和 DMax 函数或 Min 和 Max 函数。如果使用 DMin 或 DMax 函数，将在数据分组之前求最小或最大值。如果使用 Min 或 Max 函数，将在数据分组之后求字段表达式中的最小或最大值。

当需要指定条件来限制函数执行的数据范围时，可使用 DMin 或 DMax 函数。例如，要显示运到“华北”的订单的最大运费，应将文本框的“控件来源”属性设置为下列表达式：

=DMax("[Freight]", "Orders", "[ShipRegion] = 'CA'")

如果只是查找 domain 中所有记录的最小值和最大值，可使用 Min 或 Max 函数。

如果要显示的字段不在窗体所基于的记录源中，可在模块、宏或窗体上的计算控件中使用 Dmin 或 DMax 函数。


提示

虽然可以使用 DMin 或 DMax 函数来查找外部表 （外表：包含外键字段（如“客户ID”）的表，如“客户订单”。外键字段在其他表（例如“客户”）中是主键字段，并且通常处于一对多关系的“多”端。）中字段的最大值\最小值。但通过创建含有这两个表中所有需要字段的查询，然后将窗体或报表建立在这个查询的基础上，效率可能会更高。

注释  使用这些函数时，将不包括对 domain 记录未保存的更改。如果希望 DMax 或 Dmin 函数基于更改后的数据，必须先保存更改。方法是：单击“记录”菜单上的“保存记录”、将焦点移动到另一条记录，或使用 Update 方法。

示例

下面的示例返回运往英国的订单的“Freight”字段的最小和最大值。域为“Orders”表。criteria 参数将记录集的结果限制为“ShipCountry”等于 UK 的记录。

Dim curX As Currency

Dim curY As Currency

curX = DMin("[Freight]", "Orders", "[ShipCountry] = 'UK'")
curY = DMax("[Freight]", "Orders", "[ShipCountry] = 'UK'")
在下面的示例中，criteria 参数将包含“OrderDate”文本框的当前值。该文本框绑定到“Orders”表中的“OrderDate”字段。请注意，控件的两侧未加上用以表示字符串的双引号 (")。这样可以确保每次调用 DMax 函数时，Microsoft Access 都从控件中获得当前值。

Dim curX As Currency

curX = DMax("[Freight]", "Orders", "[OrderDate] = #" _

    & Forms!Orders!OrderDate & "#")
在下面的示例中，条件表达式中包含了变量 dteOrderDate。请注意，字符串表达式中使用了数字符号 (#)，这样当连接字符串时，日期将包含在 # 内。

Dim dteOrderDate As Date

Dim curX As Currency

dteOrderDate = #03/30/2000#

curX = DMin("[Freight]", "Orders", _

    "[OrderDate] = #" & dteOrderDate & "#")

DStDev、DStDevP 函数

参阅应用于示例特性

DStDev 和 DStDevP 函数用于估算特定记录集（一个域 （域：由表、查询或 SQL 表达式定义的记录集。域聚合函数返回有关特定域或记录集的统计信息。））中一组值的标准差。可以在 Visual Basic、宏 （宏：可用来使任务自动化的操作或操作集。宏在 Visual Basic for Applications 编程语言中录制。）、查询表达式、窗体或报表上的计算控件 （计算控件：在窗体、报表或数据访问页上用来显示表达式结果的控件。每当表达式所基于的值发生改变，就重新计算一次结果。）中使用 DStDev 和 DStDevP 函数。

DStDevP 函数用于计算总体，而 DStDev 函数用于计算总体抽样。

例如，可以在模块中使用 DStDev 函数来计算一组学生考试成绩的标准差。

DStDev(expr, domain, [criteria])

DStDevP(expr, domain, [criteria])

DStDev 和 DStDevP 函数具有以下参数：

	参数
	说明

	expr
	表达式，用于标识要在其中计算标准差的数值字段。它可以是用于标识表或查询中字段的字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），也可以是以该字段中的数据进行计算的表达式。在 expr 中可以包括表中字段的名称、窗体上的控件、常量或函数。如果 expr 包含函数，那么它可以是内置的，也可以是用户自定义的，但不能是另一个域聚合函数或 SQL 聚合函数。

	domain
	字符串表达式，代表组成域的记录集。可以是表名称或不需要参数的查询的查询名称。

	criteria
	可选的字符串表达式，用于限制 DStDev 和 DStDevP 函数执行的数据范围。例如，criteria 通常等价于 SQL 表达式中的 WHERE 子句，只是不含 WHERE 关键字。如果忽略 criteria，DStDev 和 DStDevP 函数将在整个域范围内计算 expr。任何包含在 criteria 中的字段必须同时也是 domain 中的字段，否则 DStDev 和 DStDevP 函数将返回 Null （Null：可以在字段中输入或用于表达式和查询，以标明丢失或未知的数据。在 Visual Basic 中，Null 关键字表示 Null 值。有些字段（如主键字段）不可以包含 Null 值。）。


说明

如果 domain 引用的记录少于两条，或满足 criteria 的记录少于两条，DStDev 和 DStDevP 函数将返回 Null，表示不能计算标准差。

无论是在宏、模块、查询表达式或者在计算控件中使用 DStDev 和 DStDevP 函数，都必须谨慎构造 criteria 参数以确保能够正确地进行计算。

DStDev 和 DStDevP 函数可用于在选择查询的“条件”行中指定条件。例如，可以在“Orders”表和“Products”表上创建一个查询来显示运输成本高于平均值加上标准差的所有产品。“Freight”字段下的条件行应包含以下表达式：

>(DStDev("[Freight]", "Orders") + DAvg("[Freight]", "Orders"))

在查询的计算字段表达式中，或更新查询 （更新查询：一种操作查询（SQL 语句），它根据所指定的条件（查找条件）更改记录集。）的“更新到”行中都可以使用 DStDev 和 DStDevP 函数。

注释  可以在总计查询 （总计查询：这种查询可针对表（一个或多个）中各不同字段显示汇总计算，如平均值或总计值。总计查询不是一个单独种类的查询，而是扩展了选择查询的灵活性。）的计算字段表达式中使用 DStDev 和 DStDevP 函数或 StDev 和 StDevP 函数。如果使用 DStDev 或 DStDevP 函数，将在分组数据之前计算值的标准差。如果使用 StDev 或 StDevP 函数，则将在分组之后计算字段表达式中值的标准差。

当需要指定条件来限制函数执行的数据范围时，可在计算控件中使用 DStDev 和 DStDevP 函数。例如，要显示运到“华北”的订单的标准差，可将文本框的“控件来源” 属性设置为以下表达式：

=DStDev("[Freight]", "Orders", "[ShipRegion] = 'CA'")

如果只是要查找 domain 中所有记录的标准差，可使用 StDev 或 StDevP 函数。


提示

如果派生 expr 的字段的数据类型为数字，DStDev 和 DStDevP 函数将返回 Double （Double 数据类型：一种基础数据类型，存储双精度浮点数。它存储 64 位数，范围大约介于 -1.797E308 到 -4.940E-324（负）之间、4.94E-324 到 1.797E308（正）之间和 0。） 数据类型。如果在计算控件中使用 DStDev 或 DStDevP 函数，可在表达式中包含一个数据类型转换函数以改进性能。

注释  使用这些函数时，将不包括对 domain 记录的未保存更改。如果希望 DStDev 或 DStDevP 函数基于更改后的数据，必须先保存更改。方法是：单击“记录”菜单上的“保存记录”、将焦点移动到另一条记录，或者使用 Update 方法。

示例

下面的示例返回运往英国的订单全体或全体抽样的标准差估计值。域为“Orders”表。criteria 参数将生成的记录集限制为“ShipCountry”值等于 UK 的记录。

Dim dblX As Double

Dim dblY As Double

' Sample estimate.

dblX = DStDev("[Freight]", "Orders", "[ShipCountry] = 'UK'")

' Population estimate.

dblY = DStDevP("[Freight]", "Orders", "[ShipCountry] = 'UK'")
下面的示例通过在 criteria 参数中使用变量 strCountry 来计算与上例相同的估计值。请注意，字符表达式中包含了单引号 (')，这样当连接字符串时，字面字符串 UK 将包含在单引号内。

Dim strCountry As String

Dim dblX As Double

Dim dblY As Double

strCountry = "UK"

dblX = DStDev("[Freight]", "Orders", _

    "[ShipCountry] = '" & strCountry & "'")

dblY = DStDevP("[Freight]", "Orders", _

    "[ShipCountry] = '" & strCountry & "'")

DSum 函数

参阅应用于示例特性

DSum 函数用于计算指定记录集（一个域 （域：由表、查询或 SQL 表达式定义的记录集。域聚合函数返回有关特定域或记录集的统计信息。））中的一组值的总和。可以在 Visual Basic、宏 （宏：可用来使任务自动化的操作或操作集。宏在 Visual Basic for Applications 编程语言中录制。）、查询表达式或计算控件 （计算控件：在窗体、报表或数据访问页上用来显示表达式结果的控件。每当表达式所基于的值发生改变，就重新计算一次结果。）中使用 DSum 函数。

例如，可以在查询的计算字段表达式中使用 DSum 函数来计算一段时期某个特定雇员的总销售额。也可以在计算控件中使用 DSum 函数来显示特定产品的动态销售总和。

DSum(expr, domain, [criteria])

DSum 函数具有以下参数：

	参数
	说明

	expr
	表达式，用于标识要对其值作总计的字段。可以是指向表或查询中字段的字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），也可以是以该字段中的数据进行计算的表达式。在 expr 中可以包括表中字段的名称、窗体上的控件、常量或函数。如果 expr 包含函数，那么它可以是内置的，也可以是用户定义的，但不能是另一个域聚合函数或 SQL 聚合函数。

	domain
	字符串表达式，代表组成域的记录集。可以是表名称或不需要参数的查询的查询名称。

	criteria
	可选的字符串表达式，用于限制 DSum 函数执行的数据范围。例如，criteria 通常等价于 SQL 表达式中的 WHERE 子句，只是不含 WHERE 关键字。如果忽略 criteria，DSum 函数将在整个域范围内计算 expr。任何包含在 criteria 中的字段必须同时也是 domain 中的字段，否则 Dsum 函数将返回 Null （Null：可以在字段中输入或用于表达式和查询，以标明丢失或未知的数据。在 Visual Basic 中，Null 关键字表示 Null 值。有些字段（如主键字段）不可以包含 Null 值。）。


说明

如果没有记录满足 criteria 参数或者域中不包含任何记录，DSum 函数将返回 Null。

无论是在宏或模块、查询表达式或者在计算控件中使用 DSum 函数，都必须谨慎构造 criteria 参数以确保能够正确地进行计算。

在查询的“条件”行、查询表达式的计算字段中或更新查询 （更新查询：一种操作查询（SQL 语句），它根据所指定的条件（查找条件）更改记录集。）的“更新到”行中，均可以使用 DSum 函数指定条件。

注释  可以在总计查询 （总计查询：这种查询可针对表（一个或多个）中各不同字段显示汇总计算，如平均值或总计值。总计查询不是一个单独种类的查询，而是扩展了选择查询的灵活性。）的计算字段表达式中使用 DSum 或 Sum 函数。如果使用 DSum 函数，将在数据分组之前计算总计值。如果使用 Sum 函数，则在分组之后计算字段表达式中值的总计值。

如果需要显示窗体或报表记录源以外的字段值的总和，可能需要使用 DSum 函数。例如，如果有一个窗体显示有关一种特定产品的信息。可以在计算控件中使用 DSum 函数来计算这种产品销售额的运行总和。


提示

如果要保持报表控件中的运行总和，当控件所基于的字段位于报表的记录源中时，可以使用控件的 RunningSum 属性，而使用DSum 函数则可保持窗体上的运行总和。

注释  使用该函数时，将不包括对 domain 记录的未保存更改。如果希望 DSum 函数基于更改后的数据，必须先保存更改。方法是：单击“记录”菜单上的“保存记录”、将焦点移动到另一条记录，或使用 Update 方法。

示例

下面的示例计算运往英国的订单的“Freight”字段的总计值。域为“Orders”表。criteria 参数将记录集的结果限制为“ShipCountry”等于 UK 的记录。

Dim curX As Currency

curX = DSum("[Freight]", "Orders", "[ShipCountry] = 'UK'")
下一个示例通过使用两个不同的条件来计算总计。注意单引号（'）和数字符号（#）需要包括在字符串表达式中，使得字符串连接时，将原义字符串包含在单引号中，将日期包含在数字符号中。

Dim curX As Currency

curX = DSum("[Freight]", "Orders", _

    "[ShipCountry] = 'UK' AND [ShippedDate] > #1-1-95#")
可以在更新查询的“更新到”行中使用域函数。例如，如果要跟踪“Products”表中各产品的当前销售情况，可以将一个名为 SalesSoFar 的新字段添加到“Products”表中，并运行一个更新查询来计算正确的值并更新记录。首先，基于“Products”表新建一个查询，再单击“查询”菜单上的“更新查询”。然后，将 SalesSoFar 字段添加到查询网格，并将下列内容输入到“更新到”行中：

DSum("[Quantity]*[UnitPrice]", "Order Details", "[ProductID] = " _

    & [ProductID])
在运行查询时，Microsoft Access 将根据“Order Details”表中的信息计算每种产品的销售总量。每种产品的销售总和将添加到“Products”表中。

DVar、DVarP 函数

参阅应用于示例特性

DVar 和 DVarP 函数用于估算特定记录集（一个域 （域：由表、查询或 SQL 表达式定义的记录集。域聚合函数返回有关特定域或记录集的统计信息。））中一组值的方差。可以在 Visual Basic、宏 （宏：可用来使任务自动化的操作或操作集。宏在 Visual Basic for Applications 编程语言中录制。）、查询表达式、窗体或报表上的计算控件 （计算控件：在窗体、报表或数据访问页上用来显示表达式结果的控件。每当表达式所基于的值发生改变，就重新计算一次结果。）中使用 DVar 和 DVarP 函数。

DVarP 函数用于计算总体的方差，而 DVar 函数用于计算总体抽样的方差。

例如，可以使用 DVar 函数来计算一组学生考试成绩的方差。

DVar(expr, domain, [criteria])

DVarP(expr, domain, [criteria])

DVar 和 DVarP 函数具有以下参数：

	参数
	说明

	expr
	字符串表达式，用于标识要在其中计算方差的数值字段。可以是用于标识表或查询中字段的字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），也可以是以该字段中的数据进行计算的表达式。在 expr 中可以包括表中字段的名称、窗体上的控件、常量或函数。如果 expr 包含函数，那么它可以是内置的，也可以是用户定义的，但不能是另一个域聚合函数或 SQL 聚合函数。

	domain
	字符串表达式，代表组成域的记录集。可以是表名称或不需要参数的查询的查询名称。

	criteria
	可选的字符串表达式，用于限制 DVar 和 DVarP 函数执行的数据范围。例如，criteria 通常等价于 SQL 表达式中的 WHERE 子句，只是不含 WHERE 关键字。如果忽略 criteria，DVar 和 DVarP 函数将在整个域范围内计算 expr。任何包含在 criteria 中的字段必须同时也是 domain 中的字段，否则 DVar 和 DVarP 函数将返回 Null （Null：可以在字段中输入或用于表达式和查询，以标明丢失或未知的数据。在 Visual Basic 中，Null 关键字表示 Null 值。有些字段（如主键字段）不可以包含 Null 值。）。


说明

如果 domain 引用的记录少于两条，或者满足 criteria 的记录少于两条，DVar 和 DVarP 函数将返回 Null，表示不能计算方差。

无论是在宏、模块、查询表达式，还是在计算控件中使用 DVar 和 DVarP 函数，都必须谨慎构造 criteria 参数以确保能够正确地进行计算。

在选择查询的“条件”行、查询中的计算字段表达式中，或更新查询的“更新到”行中均可以使用 DVar 和 DVarP 函数指定条件。

注释  可以在总计查询 （总计查询：这种查询可针对表（一个或多个）中各不同字段显示汇总计算，如平均值或总计值。总计查询不是一个单独种类的查询，而是扩展了选择查询的灵活性。）的计算字段表达式中使用 DVar 和 DVarP 函数或 Var 和 VarP 函数。如果使用 DVar 或 DVarP 函数，将在数据分组之前计算方差。如果使用 Var 或 VarP 函数，则在分组之后计算字段表达式中值的方差。

当需要指定 criteria 来限制函数执行的数据范围时，可在计算控件中使用 DVar 和 DVarP 函数。例如，要显示运到 California 的订单的方差，可将文本框的“控件来源”属性设为如下表达式：

=DVar("[Freight]", "Orders", "[ShipRegion] = 'CA'")

如果只是在 domain 的所有记录中查找标准差，可使用 Var 或 VarP 函数。

注释  在使用该函数时，将不包括对 domain 记录的未保存更改。如果希望 DVar 或 DVarP 函数基于更改后的数据，必须先保存更改。方法是：单击“记录”菜单上的“保存记录”、将焦点移动到另一条记录，或使用 Update 方法。

示例

下面的示例返回运往英国的订单总体和总体抽样的方差估算值。域是“Orders”表。criteria 参数将记录集的结果限制为“ShipCountry”等于 UK 的记录。

Dim dblX As Double

Dim dblY As Double

' Sample estimate.

dblX = DVar("[Freight]", "Orders", "[ShipCountry] = 'UK'")

' Population estimate.

dblY = DVarP("[Freight]", "Orders", "[ShipCountry] = 'UK'")
下一个示例在 criteria 参数中使用一个变量 strCountry 来返回估算值。请注意单引号（'）在字符串表达式中使用，使得字符串连接时，将字面字符串“UK”包含在单引号中。

Dim strCountry As String|

Dim dblX As Double

strCountry = "UK"

dblX = DVar("[Freight]", "Orders", "[ShipCountry] = '" _

    & strCountry & "'")

EuroConvert 函数

参阅应用于示例特性

如果要将数字转换为欧元或者将欧元转换为某个参与货币，可以使用 EuroConvert 函数。使用该函数还可以将数字从一个参与货币转换为使用欧元作为中介（三方）的其他货币格式。EuroConvert 函数使用由欧盟建立的固定转换汇率。

EuroConvert(number, sourcecurrency, targetcurrency, [fullprecision, triangulationprecision])

	参数
	说明

	number
	需要转换的数字，或者对包含数字的字段的引用。

	sourcecurrency
	字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。），或者对包含字符串的字段的引用，对应着与要转换的货币的“国际标准化组织”(ISO) 的缩写。下表列出了可能出现的 ISO 代码。

	
	货币
	ISO 代码
	计算精度
	显示精度

	
	比利时法郎
	BEF
	0
	0

	
	卢森堡法郎
	LUF
	0
	0

	
	德国马克
	DEM
	2
	2

	
	西班牙比塞塔
	ESP
	0
	0

	
	法国法郎
	FRF
	2
	2

	
	爱尔兰镑
	IEP
	2
	2

	
	意大利里拉
	ITL
	0
	0

	
	荷兰盾
	NLG
	2
	2

	
	奥地利先令
	ATS
	2
	2

	
	葡萄牙埃斯库多
	PTE
	0
	0

	
	芬兰马克
	FIM
	2
	2

	
	欧元
	EUR
	2
	2

	
	在前面的表中，计算精度根据转换货币决定要将结果四舍五入到哪一个货币单位。例如，当转换德国马克时，计算精度为 2，结果将四舍五入到最接近的芬尼，100 芬尼等于一马克。显示精度决定了在包含结果的字段中显示的小数位数。
	
	
	

	
	EuroConvert 函数的后续版本可能支持其他货币。有关新参与货币的信息和对 EuroConvert 函数的更新，请参阅 Microsoft Office 欧元货币网站。
	
	
	

	
	货币
	ISO 代码
	
	

	
	丹麦克朗
	DKK
	
	

	
	德拉克马
	GRD
	
	

	
	瑞典克朗
	SEK
	
	

	
	英磅
	GBP
	
	

	targetcurrency
	字符串表达式，或对包含字符串的字段的引用，对应着与要将数字转换到的货币的 ISO 代码。有关 ISO 代码的列表，请参阅 sourcecurrency 参数的说明。
	
	
	

	fullprecision
	可选项。Boolean （布尔值：取值为 true（非零）或 false (0) 的表达式。可使用关键字 True 和 False 分别提供值 -1 和 0。字段数据类型 Yes/No 是布尔型，而且 Yes 的值为 -1。） 值。该值为 True (1) 时，忽略与货币相关的四舍五入规则 （在 sourcecurrency 参数说明中也称为显示精度）并使用 6 个有效数字转换因子，不向上进位；该值为 False (0) 时，使用与货币相关的四舍五入规则显示结果。如果省略该参数，默认值为 False。
	
	
	

	triangulationprecision
	可选项。大于或等于 3 的 Integer （Integer 数据类型：一种用来存储整数的基本数据类型。Integer 变量是以 16 位（2 字节）数字存储的，数值范围从 -32,768 到 32,767。） 值，该值指定当以欧元值作为换算两个国家/地区的货币的中介时，计算精度的有效数字位数。
	
	
	


说明

所有末尾的零都会被截断，无效参数都返回 #Error。

如果源 ISO 代码与目标 ISO 代码一样，则数字的原始值仍然是活动的。

该函数不应用格式。

EuroConvert 函数使用由欧盟建立的当前汇率。如果汇率发生变化，Microsoft 将更新该函数。若要获得有关当前有效的规则和汇率的全部信息，请参阅欧洲委员会有关欧元的出版物。有关获得这些出版物的信息，请参阅 Microsoft Office 欧元货币网站。

示例

第一个示例将 1.20 德国马克转换为欧元对应的值（结果为 0.61）。第二个示例将 1.47 法国法郎转换为德国马克（结果为 0.44 DM）。此处假设转换汇率为 1 欧元等于 6.55858 法国法郎或者等于 1.92974 德国马克。

EuroConvert(1.20,"DEM","EUR")

EuroConvert(1.47,"FRF","DEM",TRUE,3)

Eval 函数

参阅应用于示例特性

使用 Eval 函数可以计算一个结果为文本字符串或数值的表达式 （表达式：运算符、字段名、函数、文字和常量的组合，可得出单个结果。表达式可指定条件（例如，Order Amount>10000）或在字段值上执行计算（例如，Price*Quantity）。）的值。

可以构造一个字符串，然后就像是一个真正的表达式一样把字符串传给 Eval 函数，。Eval 函数将计算字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。）并返回计算结果值。例如， Eval("1 + 1") 返回 2。

如果传给 Eval 函数的字符串包含一个函数的名称，Eval 函数会返回该函数的返回值。例如，Eval("Chr$(65)") 返回 "A"。

Eval(stringexpr)

stringexpr 参数是一个对字母数字文本字符串进行计算的表达式。例如，stringexpr 可以是一个返回字符串或数值的函数，或是对窗体上的控件 （控件：允许用户控制程序的图形用户界面对象，如文本框、复选框、滚动条或命令按钮等。可使用控件显示数据或选项、执行操作或使用户界面更易阅读。）的引用。stringexpr 参数的值必须对字符串或数值进行计算，但不能对 Microsoft Access 对象 （Microsoft Access 对象：由 Access 定义的一种对象，它与 Access、Access 界面或应用程序的窗体、报表和数据访问页相关；而且，可以用来对输入和显示数据所采用的界面的元素进行编程。）进行计算。

注释  如果把函数名传给 Eval 函数，则必须在 stringexpr 参数中的函数名后加上括号。例如：

' ShowNames is user-defined function.

Debug.Print Eval("ShowNames()")   

Debug.Print Eval("StrComp(""Joe"",""joe"", 1)")

Debug.Print Eval("Date()")

说明

可以在窗体或报表上的计算控件 （计算控件：在窗体、报表或数据访问页上用来显示表达式结果的控件。每当表达式所基于的值发生改变，就重新计算一次结果。）中，或者宏或模块中使用 Eval 函数。Eval 函数将返回一个字符串或数值类型的 Variant （Variant 数据类型：变量的默认数据类型，如果 Def type 语句没有生效，则该数据类型没有类型声明字符。Variant 可以存储数值、字符串、日期/时间、Null 或 Empty 数据。）。

stringexpr 参数必须是一个以字符串形式存储的表达式。如果传给 Eval 函数的字符串不包括数值表达式或函数名称而仅仅是一个简单的文本字符串，则会出现运行时错误 （运行时错误：只有当应用程序运行时才能检测到的错误。）。例如， Eval("Smith") 将产生错误。

使用 Eval 函数可决定存储在控件中的 Value 的属性值。下面的示例传给 Eval 函数一个包含对控件的完整引用的字符串，然后在对话框中显示该控件的当前值。

Dim ctl As Control

Dim strCtl As String

Set ctl = Forms!Employees!LastName

strCtl = "Forms!Employees!LastName"

MsgBox ("The current value of " & ctl.Name & " is " & Eval(strCtl))

使用 Eval 函数可以访问在 Visual Basic 中通常不能使用的表达式运算符。例如，不能在代码中直接使用 SQL 运算符 Between...And 或 In，但是在传给 Eval 函数的表达式中可以使用它们。

下面的示例用于确定“订单”窗体上的“货主地区”控件的值是否为几个特定的州缩写名称中的一个。如果字段中包含了其中一个缩写名称， intState 的值将为 True（-1）。请注意，在示例中使用了单引号 (') 以在字符串中包含另一个字符串。

Dim intState As Integer

intState = Eval("Forms!Orders!ShipRegion In " _

    & "('AK', 'CA', 'ID', 'WA', 'MT', 'NM', 'OR')")

示例

下面的示例中，假设有 50 个定义为 A1、 A2 这样的一系列函数，然后使用 Eval 函数依次调用每个函数。

Sub CallSeries()

    Dim intI As Integer

    For intI = 1 To 50

        Eval("A" & intI & "()")
    Next intI

End Sub

下面的示例将触发一个 Click 事件，就像用户单击窗体上的一个按钮一样。如果按钮的 OnClick 属性值以等号（=）打头表明它是函数名，Eval 函数就会调用该函数，这与触发 Click 事件的效果相同。如果属性值不以等号打头，那么它必须指定一个宏的名称。DoCmd 对象的 RunMacro 方法运行该宏。

Dim ctl As Control

Dim varTemp As Variant

Set ctl = Forms!Contacts!HelpButton

If (Left(ctl.OnClick, 1) = "=") Then

    varTemp = Eval(Mid(ctl.OnClick,2))
Else

    DoCmd.RunMacro ctl.OnClick

End If

不同字符串值的不同格式（Format 函数）


在 Microsoft Access 1.x 和 2.0 版中，可以使用 Format 函数为零长度字符串 （零长度字符串：不含字符的字符串。可以使用零长度字符串来表明您知道该字段没有值。输入零长度字符串的方法是键入两个彼此之间没有空格的双引号 (" ")。）返回一个值而为 Null （Null：可以在字段中输入或用于表达式和查询，以标明丢失或未知的数据。在 Visual Basic 中，Null 关键字表示 Null 值。有些字段（如主键字段）不可以包含 Null 值。） 值返回另一个值。例如，可以使用如下带有 Format 函数的格式表达式从代码中返回适当的字符串值：

Dim varX As Variant

Dim varStrX As Variant

' Assign some value to varStrX and pass to Format function.

varX = Format(varStrX, "@;ZLS;Null")

在 Microsoft Access 97 及更高版本中，必须单独检测 Null 的情况，然后根据结果返回适当的值。例如，在含 Format 函数的表达式中可以象下面这样使用 IIF 函数：

varX = IIf(IsNull(varStrX),"Null", Format(varStrX, "@;ZLS"))

只有在根据字符串是零长度字符串还是 Null 值来使用 Format 函数设置字符串格式时，这种更改才会应用。与 Format 函数一起使用的其他格式表达式的用法与以前版本相同。

如果要将数据库从 Microsoft Access 1.x 或 2.0 转换为 Microsoft Access 2002 或更高版本，则需要更改代码来单独测试 Null 值的情况。

GUIDFromString 函数

参阅应用于示例特性

GUIDFromString 函数将字符串 （字符串：字符集合，能包含数字和文字。）转换为 GUID （GUID：在 Access 数据库中，一种用于建立同步复制唯一标识符的 16 字节字段。GUID 用于标识副本、副本集、表、记录和其他对象。在 Access 数据库中，GUID 是指同步复制 ID。），其值为 Byte （“字节”数据类型：Access 数据库中用来存储较小正整数的数据类型，取值范围从 0 至 255。） 数据类型的数组 （数组：用于建立可生成多个结果或可对在行和列中排列的一组参数进行运算的单个公式。数组区域共用一个公式；数组常量是用作参数的一组常量。）。

GUIDFromString(stringexpression)

GUIDFromString 函数具有以下参数：

	参数
	说明

	stringexpression
	对字符串形式的 GUID 进行计算的字符串表达式 （字符串表达式：任一求值为一列连续字符的表达式。表达式的元素可以是：返回字符串或字符串 Variant (VarType 8) 的函数；字符串字面值、常量、变量或 Variant。）。


说明

Microsoft Jet 数据库引擎 （Microsoft Jet 数据库引擎：Access 数据库系统的一部分，可以在用户和系统数据库中检索和存储数据。可以将其视为构建数据库系统（如 Access）之基础的数据管理器。）将 GUID 保存为 Byte 类型的数组。但是，Microsoft Access 不能从窗体 （窗体：Access 数据库对象之一，可以在这种对象上放置控件，用于执行操作，或在字段中输入、显示、编辑数据。）或报表 （报表：Access 数据库对象的一种，用于按照您的规格打印设定格式且组织好的信息。报表的示例有销售额汇总、电话列表以及邮件标签。）上的控件 （控件：允许用户控制程序的图形用户界面对象，如文本框、复选框、滚动条或命令按钮等。可使用控件显示数据或选项、执行操作或使用户界面更易阅读。）中返回 Byte 数据。为了从控件中返回 GUID 的值，必须将它转换为字符串。若要将 GUID 转换为字符串，请使用 StringFromGUID 函数。若要将字符串转换为 GUID，请使用 GUIDFromString 函数。

示例

下面的示例使用 GUIDFromString 函数将字符串转换为 GUID。该字符串是以字符串形式存储在同步复制的“雇员”表中的 GUID。字段 s_GUID 是隐藏字段，该字段将添加到同步复制数据库中的每个同步复制表中。

Sub CheckGUIDType()

    Dim dbsConn As ADODB.Connection

    Dim rstEmployees As ADODB.Recordset

    ' Make a connection to the current database.

    Set dbsConn = Application.CurrentProject.Connection

    Set rstEmployees = New ADODB.Recordset

    rstEmployees.Open "Employees", dbsConn, , , adCmdTable

    ' Print the GUID to the immediate window.

    Debug.Print rst!s_GUID

    Debug.Print TypeName(rst!s_GUID)

    Debug.Print TypeName(GuidFromString(rst!s_GUID))

    Set rstEmployees = Nothing

    Set dbsConn = Nothing

End Sub

Nz 函数

参阅应用于示例特性

当一个 Variant （Variant 数据类型：变量的默认数据类型，如果 Def type 语句没有生效，则该数据类型没有类型声明字符。Variant 可以存储数值、字符串、日期/时间、Null 或 Empty 数据。） 为 Null （Null：可以在字段中输入或用于表达式和查询，以标明丢失或未知的数据。在 Visual Basic 中，Null 关键字表示 Null 值。有些字段（如主键字段）不可以包含 Null 值。） 时，使用 Nz 函数可以返回 0、零长度字符串 （零长度字符串：不含字符的字符串。可以使用零长度字符串来表明您知道该字段没有值。输入零长度字符串的方法是键入两个彼此之间没有空格的双引号 (" ")。） (" ") 或其他指定值。例如，可以使用该函数将 Null 值转换为其他值并防止它通过表达式扩散。

Nz(variant, [valueifnull])

Nz 函数具有以下参数：

	参数
	说明

	variant
	数据类型 （数据类型：决定字段可拥有的数据类型的字段特征。数据类型包括 Boolean、Integer、Long、Currency、Single、Double、Date、String 和 Variant（默认）。）为 Variant 的变量。

	Valueifnull
	可选参数（除非在查询中使用）。一个 Variant，提供当 variant 参数为 Null 时的返回值。该参数使您能够返回一个除 0 或零长度字符串以外的其他值。 

注释  如果在查询的表达式中使用不带 ValueIfNull 参数的 Nz 函数，则对包含空值的字段，其返回结果为零长度字符串。


如果 variant 参数的值为 Null，Nz 函数将返回零或零长度字符串（当用在查询表达式中时，总是返回零长度字符串），这取决于上下文表明该值是数值还是字符串。如果包含了可选 valueifnull 参数，则当 variant 参数为 Null 时，Nz 函数将返回由该参数指定的值。当在查询表达式中使用时，NZ 函数总是包含 valueifnull 参数。

如果 variant 的值不为 Null，则 Nz 函数返回 variant 的值。

说明

Nz 函数对可能包含 Null 值的表达式是很有用的。要使表达式即使在包含 Null 值时也能计算得到一个非 Null 值，可使用 Nz 函数来返回 0、零长度字符串或一个自定义的返回值。

例如，表达式 2 + varX 在 Variant varX 为 Null 时总是返回 Null 值。而 2 + Nz(varX) 却返回 2。

可以经常使用 Nz 函数作为 IIf 函数的替代函数。例如，在下面的代码中，需要用两个包含 IIf 函数的表达式来返回所需的结果。第一个包含 IIf 函数的表达式用于检查变量的值，并且如果该值为 Null，则将其转换为零。

varTemp = IIf(IsNull(varFreight), 0, varFreight)

varResult = IIf(varTemp > 50, "High", "Low")

在下一个示例中，Nz 函数提供与第一个表达式相同的功能，并且一步即可获得所需的值而不需要分两步进行。

varResult = IIf(Nz(varFreight) > 50, "High", "Low")

如果为可选参数 valueifnul 提供了一个值，则当 variant 为 Null 时，将返回该值。通过包含该可选参数，可以避免使用含有 IIf 函数的表达式。例如，下面的表达式使用 IIf 函数，在 varFreight 的值为 Null 时返回一个字符串。

varResult = IIf(IsNull(varFreight), "No Freight Charge", varFreight)

在下一个示例中，为 Nz 函数提供的可选参数提供在 varFreight 为 Null 时返回的字符串。

varResult = Nz(varFreight, "No Freight Charge")

示例

下面的示例对窗体上的一个控件进行计算，并根据控件值返回两个字符串之一。如果控件值为 Null，过程就会使用 Nz 函数将 Null 值转换为零长度字符串。

Public Sub CheckValue()

    Dim frm As Form

    Dim ctl As Control

    Dim varResult As Variant

    ' Return Form object variable pointing to Orders form.

    Set frm = Forms!Orders

    ' Return Control object variable pointing to ShipRegion.

    Set ctl = frm!ShipRegion

    ' Choose result based on value of control.

    varResult = IIf(Nz(ctl.Value) = vbNullString, _

        "No value.", "Value is " & ctl.Value & ".")

    ' Display result.

    MsgBox varResult, vbExclamation

End Sub

StringFromGUID 函数

参阅应用于示例特性

StringFromGUID 函数将值为 Byte （“字节”数据类型：Access 数据库中用来存储较小正整数的数据类型，取值范围从 0 至 255。） 数据类型的数组 （数组：用于建立可生成多个结果或可对在行和列中排列的一组参数进行运算的单个公式。数组区域共用一个公式；数组常量是用作参数的一组常量。）的 GUID （GUID：在 Access 数据库中，一种用于建立同步复制唯一标识符的 16 字节字段。GUID 用于标识副本、副本集、表、记录和其他对象。在 Access 数据库中，GUID 是指同步复制 ID。） 转换为字符串 （字符串：字符集合，能包含数字和文字。）。

StringFromGUID(guid)

StringFromGUID 函数具有以下参数：

	参数
	说明

	guid
	Byte 数据类型的数组，用于唯一地标识操作系统中的应用程序、组件或数据项。


说明

Microsoft Jet 数据库引擎 （Microsoft Jet 数据库引擎：Access 数据库系统的一部分，可以在用户和系统数据库中检索和存储数据。可以将其视为构建数据库系统（如 Access）之基础的数据管理器。）将 GUID 存储为 Byte 类型的数组。但是，Microsoft Access 不能由窗体 （窗体：Access 数据库对象之一，可以在这种对象上放置控件，用于执行操作，或在字段中输入、显示、编辑数据。）或报表 （报表：Access 数据库对象的一种，用于按照您的规格打印设定格式且组织好的信息。报表的示例有销售额汇总、电话列表以及邮件标签。）上的控件 （控件：允许用户控制程序的图形用户界面对象，如文本框、复选框、滚动条或命令按钮等。可使用控件显示数据或选项、执行操作或使用户界面更易阅读。）返回 Byte 数据。为了从控件返回 GUID 的值，必须将它转换为字符串。若要将 GUID 转换为字符串，请使用 StringFromGUID 函数。若要将字符串转换为 GUID，请使用 GUIDFromString 函数。

例如，在进行数据库的同步复制时，可能需要引用包含 GUID 的字段。若要返回窗体上与包含 GUID 的字段绑定的控件的值，请使用 StringFromGUID 函数将 GUID 转换为字符串。

请注意，为将控件与同步复制表的 s_GUID 字段绑定，必须在“选项”对话框（通过“工具”菜单打开）的“视图”选项卡上选中“系统对象”复选框。

示例

下面的示例以字符串形式返回“雇员”窗体中 s_GUID 控件的值，然后将其指定给字符串参数。s_GUID 控件将绑定到 s_GUID 字段，该字段为系统字段之一，将添加到同步复制的数据库中的每个同步复制表中。

Public Sub StringValueOfGUID()

    Dim ctl As Control

    Dim strGUID As String

    ' Get the GUID.

    Set ctl = Forms!Employees!s_GUID

    Debug.Print TypeName(ctl.Value)

    ' Convert the GUID to a string.

    strGUID = StringFromGUID(ctl.Value)
    Debug.Print TypeName(strGUID)

End Sub

